



UNIVERSITY OF NEVADA-RENO

Nevada Bureau of Mines and Geology
University of Nevada-Reno
Reno, Nevada 89557-0088
(702) 784-6691

1988

NBMG OPEN FILE REPORT 88-3

PETCAL (version 2.1)
AN EXPANDED AND PERSONAL COMPUTER COMPATIBLE
BASIC LANGUAGE PROGRAM FOR PETROLOGIC
CALCULATIONS

Donald M. Hudson

This information should be considered preliminary.
It has not been edited or checked for completeness
or accuracy.

[Redacted area]

PETCAL (version 2.1)
An Expanded and Personal Computer Compatible
BASIC Language Program for Petrologic Calculations

Donald M. Hudson
1988

Nevada Bureau of Mines and Geology

Open File Report 88-3

INTRODUCTION

The original PETCAL (PETrologic CALculations) program (Bingler and others, 1976) was published before the advent of widespread use of personal computers. This version of PETCAL is adapted for use with IBM-compatible personal computers using BASIC (Beginners All-purpose Symbolic Instruction Code) language. The program has been rewritten to simplify programming logic so the program may be expanded by the user to incorporate additional petrologic manipulations. The program is also designed to be easy to use with only a minimum of experience with personal computers and is interactive in a simple manner. Various short-form BASIC commands have been avoided to make use of the program compatible with as many systems as possible and with various versions of BASIC.

Required to use the program are an IBM-compatible computer with a BASIC language compiler and an on-line printer. The PETCAL program and its companion PETDAT program are available on diskette from the Nevada Bureau of Mines and Geology for a modest cost.

This program is an extension of PETCAL (Bingler and others, 1976). This report gives additional information and serves as a supplement. Since the program has been rewritten and extended, a full listing of the present program is given in Appendix A and a sample output listing is given in Appendix B. Many petrologic indices, as additions to those in the original PETCAL (Bingler and others, 1976), have been added to the program and explained in Section 2 of this report.

Also included in this report is a short BASIC program to create a datafile for PETCAL called PETDAT (PETrologic DATA) that is presented in Appendix C. The PETDAT program offers the flexibility of creating permanent datafiles for PETCAL which can be stored for future use and be quickly accessed.

SECTION 1. USING PETCAL

After loading the program under the BASIC language, be sure the printer is on and the "CAPS LOCK" is activated on the keyboard. The program is then ready to run.

The first statement to appear on the CRT will be:

IS DATA IN SEPARATE DATAFILE (Y/N) ?

Since the program is written to accept input from a separate datafile or one sample at a time, a "N" response means that data is to be entered one sample at a time and the next program step will be skipped; a "Y" response requires that a preexisting datafile has been created (see the PETDAT program in Appendix C) and will be called into the program in the next step. If the previous answer was "Y" then the program will respond with:

ENTER DATAFILE NAME ?

Enter the name of the datafile to be called into the program. If the data is being called from a datafile, the program will call for the next two responses below. After the second question has been answered, the program will proceed through the datafile uninterrupted until output to the printer is completed. The message "Input past end in 350" indicates that output is complete.

Whether or not the data is to be called from a separate datafile, the next prompt will be:

SHALL I USE NORMALIZED OXIDE VALUES (Y/N) ?

This option is used throughout the program and can only be changed by exiting the program and restarting. If "Y" then all input values will be normalized to total 100 percent.

The next prompt is:

SHALL I RECALCULATE FE2O3-FEO (Y/N) ?

This option is designed for use when total iron is given as either Fe_2O_3 or as FeO . A "Y" response will recalculate the iron throughout the entire program and can only be changed by exiting the program and then restarting. The iron values are recalculated using the formula:

$$\% \text{Fe}_2\text{O}_3 = \% \text{TiO}_2 + 1.5$$

with the "excess converted to FeO , as suggested by Irvine and Baragar (1971). For basaltic rocks ($\text{SiO}_2 < 53\%$) the ratio is adjusted so that:

$$\text{Fe}_2\text{O}_3 / (\text{Fe}_2\text{O}_3 + \text{FeO}) = 0.2$$

a value typically not exceeded by unoxidized modern basalts (Hughes and Hussey, 1976).

If the data is not being called from a separate datafile the program will then prompt:

ENTER SAMPLE NUMBER ?

Enter the sample number (name) in any combination of characters and of any length.

The next series of prompts will be for the input data in weight percent, one value at a time. WARNING: a value other than zero must be entered for each oxide except P₂O₅ and MnO, otherwise division by zero may occur, an error message will appear, and the program will terminate. A value of zero may be entered for either Fe₂O₃ or FeO if the iron values are to be recalculated. Then series of prompts will appear as follows:

INPUT SIO2	?
INPUT AL2O3	?
INPUT FE2O3	?
INPUT FEO	?
INPUT MGO	?
INPUT CAO	?
INPUT NA2O	?
INPUT K2O	?
INPUT TiO2	?
INPUT P2O5	?
INPUT MNO	?

The program will then ask:

IS THE DATA CORRECT (Y/N) ?

If the data is incorrect, then enter "N" and reenter all of the data. If the data is correct, enter "Y" and the program will print output to the printer. A sample output is given in Appendix B. After output is complete for that sample, the program will ask:

ANOTHER SAMPLE (Y/N) ?

If "N" is entered, the program will be exited. If "Y" is entered, the program will recycle to:

ENTER SAMPLE NUMBER ?

PETROLOGIC INDICES

The following petrologic indices have been added to the PETCAL program. For other petrologic indices generated by the program see Bingler and others (1976).

FELSIC-MAFIC INDEX

The felsic-mafic index, originally used by Segerstrom and Young (1972, p. 34-35), is defined as:

$$\frac{\text{SiO}_2 + \text{Na}_2\text{O} + \text{K}_2\text{O}}{\text{FeO} + \text{Fe}_2\text{O}_3 + \text{MgO} + \text{CaO}}$$

The index can be used in classifying rocks as in Table 1.

Table 1. Felsic-mafic index classification

Rock type	Felsic-mafic index ¹	Felsic-mafic index ²
Extreme alkali granite (rhyolite)	>50	
Alkali granite (rhyolite)	25-50	>25
Granite (rhyolite)	15-25	15-25
Quartz monzonite (quartz latite)	10-15	10-15
Granodiorite (rhyodacite)	7-10	7-10
Quartz diorite (dacite)	5- 7	
Quartz diorite (quartz andesite)		5- 7
Monzonite (latite)		3- 5
Monzonite (latite) K ₂ O>3%	3- 5	
Trachyandesite K ₂ O<3%	3- 5	
Diorite (andesite)	2.1- 3	2- 3
Gabbro (basalt)	1.4-2.1	1- 2
Ultramafics	<1.4	<1

¹ Modification used in Bonham and Garside (1979).

² Original classification of Segerstrom and Young (1972).

WEATHERING INDEX

A chemical index for weathering of silicate rocks was proposed by Parker (1970) based upon susceptibility of different elements to weathering. The index is defined by the following expression

$$\frac{(\text{Na})_a}{0.35} + \frac{(\text{Mg})_a}{0.9} + \frac{(\text{K})_a}{0.25} + \frac{(\text{Ca})_a}{0.7} \times 100$$

where (X)_a is the atomic proportion of element X defined as atomic percentage divided by atomic weight. The denominator is the bond strength of X with oxygen. Table 2 gives examples of the use of the weathering index.

Table 2. Application of the weathering index to chemical analyses of profiles. From Parker (1970).

Profile	Values of the Index						Rock Type
	Fresh rock			Weathered horizons			
S-1	78.2	71.8	68.3	43.3	34.6	30.3	granite
S-2	76.0	77.5	68.5	52.7			granite
T-1	82.7	76.0	75.2	62.9	52.9	49.5	granite
T-2	80.0	76.3	74.7	70.8	43.4	47.2	granite
E-1	75.9	65.2	67.2	69.7	37.7	22.4	granodiorite
Windy	72.8	43.7	20.3	13.4	8.8		hypersthene-andesite
Cohasset	70.0	50.9	6.7	2.3	2.1		olivine-andesite
Kiama	93.5	62.8	30.4	20.4	15.7	19.8	latite
Bathurst	87.7	64.7	4.7	4.5	4.5	4.5	basalt
Inverell	81.9	13.8	11.9	6.8	6.0	3.1	basalt
Casino	106.0	85.4	45.5	30.3	15.8	9.7	basalt
Guya	89.4	78.0	29.6	27.6	19.7	18.8	basalt
Murrundi	87.6	60.6	62.3	32.5	46.8	35.0	basalt
							13.6

ALTERATION INDEX

Ishikawa and others (1976) proposed an index for alteration of felsic rocks which is defined as the following expression (in weight percent):

$$\frac{\text{MgO} + \text{K}_2\text{O}}{\text{MgO} + \text{CaO} + \text{Na}_2\text{O} + \text{K}_2\text{O}} \times 100$$

Izawa and others (1978) suggest a value of 50 being an upper threshold for unaltered felsic rocks but must be used with caution. A high value indicates magnesium enrichment and alkali depletion.

PERALUMINOUS INDEX

The peraluminous index is defined by the following expression (in moles):

$$\frac{\text{Al}_2\text{O}_3 - (\text{CaO} + \text{Na}_2\text{O} + \text{K}_2\text{O})}{\text{Al}_2\text{O}_3} \times 100$$

PERALKALINE INDEX

The peralkaline index is defined by the following expression (in moles):

$$\frac{\text{Na}_2\text{O} + \text{K}_2\text{O}}{\text{Al}_2\text{O}_3}$$

THETA INDEX

The theta index was proposed by Sugimura (1968) as an indication of parental magma type. The index is defined by the following expression (in weight percent):

$$\Theta = \text{SiO}_2 - 47 \frac{\text{Na}_2\text{O} + \text{K}_2\text{O}}{\text{Al}_2\text{O}_3}$$

SUITE INDEX

Similar to the theta index, the suite index was proposed by Rittman (1962) as an indication of parental magma type and is almost constant for any one volcanic suite. The index is defined by the following expression (in weight percent):

$$\sigma = \frac{(\text{Na}_2\text{O} + \text{K}_2\text{O})^2}{\text{SiO}_2 - 43}$$

COMMONLY USED MAGMATIC SERIES

This portion of the PETCAL program classifies the rock using a number of commonly used magmatic series. The user is cautioned to consult the cited references for a complete understanding of the classifications presented.

ALUMINA SATURATION

A commonly used classification scheme for alumina saturation was proposed by Shand (1943). The following is the classification (using mole numbers):

peraluminous: $\text{Al}_2\text{O}_3 > \text{Na}_2\text{O} + \text{K}_2\text{O} + \text{CaO}$

metaluminous: $\text{Al}_2\text{O}_3 > \text{Na}_2\text{O} + \text{K}_2\text{O}$ but $< \text{Na}_2\text{O} + \text{K}_2\text{O} + \text{CaO}$

subaluminous: Al_2O_3 about $= \text{Na}_2\text{O} + \text{K}_2\text{O}$ (± 0.5)

peralkaline: $\text{Al}_2\text{O}_3 < \text{Na}_2\text{O} + \text{K}_2\text{O}$

IRVINE AND BARAGAR'S CLASSIFICATION

Irvine and Baragar (1971) proposed classification of igneous suites into alkalic and subalkalic on the basis of a $\text{Na}_2\text{O} + \text{K}_2\text{O}$ versus SiO_2 plot (fig. 1). They further subdivided the subalkalic series into tholeiitic and calc-alkalic series based on a plot of Al_2O_3 versus normative anorthorthite composition (fig. 2).

MIYASHIRO'S CLASSIFICATION

Miyashiro (1974) proposed a classification for non-alkalic series rocks on the basis of a plot of SiO_2 versus FeO^*/MgO (FeO^* is total Fe calculated as FeO). Figure 3 shows the fields defined by Miyashiro.

KUNO'S CLASSIFICATION

Kuno (1968) proposed a means of classification of parental magmas on the basis of plotting $\text{Na}_2\text{O} + \text{K}_2\text{O}$ versus




Figure 1. Alkali-silica diagram for alkalic and subalkalic rock series of Irvine and Baragar (1971).




Figure 2. Alumina-normative anorthite composition diagram for calc-alkalic and tholeiitic rock series of Irvine and Baragar (1971).




Figure 3. Silica- FeO^*/MgO diagram for calc-alkalic and tholeiitic rock series of Miyashiro (1974).




Figure 4. Alkali-silica diagram for tholeiitic, high alumina, and alkalic rock series of Kuno (1968).

SiO_2 (fig. 4). The boundaries of each field are defined by irregular curves that are approximated in the PETCAL program by straight line segments.

MACDONALD AND KATSURA'S CLASSIFICATION

MacDonald and Katsura (1964) defined two parental magma series on the basis of a plot of $\text{K}_2\text{O} + \text{Na}_2\text{O}$ versus SiO_2 (fig. 5).

CHAPPELL AND WHITE'S CLASSIFICATION

Chappell and White (1974) proposed the terms I-type (igneous) and S-type (sedimentary) to indicate the source of igneous melts. They used the expression (in moles):

$$\text{Al}_2\text{O}_3 / (\text{Na}_2\text{O} + \text{K}_2\text{O} + \text{CaO})$$

as the major parameter to define the two suites. If the ratio is >1.1 then the rock is likely to be S-type; if <1.1 then the rock is likely to be I-type. Since other criteria are used, this classification generated by PETCAL should be applied with caution.

IUGS VOLCANIC ROCK NAME

The IUGS Subcommission on the Systematics of Igneous Rocks has proposed a chemical classification of volcanic rocks (Zanettin, 1984). The classification is based upon a plot of normalized weight percent $\text{Na}_2\text{O} + \text{K}_2\text{O}$ versus SiO_2 (fig 6). Only the root names are generated by the PETCAL program without distinction as to sodic or potassic types.




Figure 5. Alkali-silica diagram for alkali-olivine and tholeiitic rock series of MacDonald and Katsura (1964).




Figure 6. Alkali-silica diagram showing root names for chemical classification of volcanic rocks (Zanettin, 1984, Fig. 1).

REFERENCES

- Bingler, E.C., Trexler, D.T., Kemp, W.R., and Bonham, H.F., Jr., 1976, PETCAL: a BASIC language computer program for petrologic calculations: Nevada Bureau of Mines and Geology Report 28, 27 p.
- Bonham, H.F., Jr., and Garside, L.J., 1979, Geology of the Tonopah, Lone Mountain, Klondike, and northern Mud Lake quadrangles, Nevada: Nevada Bureau of Mines and Geology Bulletin 92, 142 p.
- Chappell, B.W., and White, A.J.R., 1974, Two contrasting granite types: Pacific Geology, v. 8, p. 173-174.
- Hughes, C.J., and Hussey, E.M., 1976, M and Mg values in igneous rocks: proposed usage and a comment on currently employed Fe_2O_3 corrections: *Geochimica et Cosmochimica Acta*, v. 40, p. 485-486.
- Irvine, T.N., and Baragar, W.R.A., 1971, A guide to the chemical classification of the common volcanic rocks: *Canadian Journal Of Earth Sciences*, v. 8, p. 523-548.
- Ishikawa, Y., Sawaguchi, T., Iwaya, S., and Horiuchi, M., 1976, Delineation of prospecting targets for Kuroko deposits based on modes of volcanism of underlying dacite and alteration haloes: *Mining Geology*, v. 26, p. 105-117 (in Japanese).
- Izawa, E., Yoshida, T., and Saito, R., 1978, Geochemical characteristics of hydrothermal alteration around the Fukazawa Kuroko deposit, Japan: *Mining Geology*, v. 28, p. 325-335.
- Kuno, H., 1968, Differentiation of basalt magmas, in Hess, H.H., and Poldervaart, A., ed., *Basalts: The Poldervaart Treatise on Rocks of Basaltic Composition*: Interscience Publishers, v. 2, p. 623-688.
- MacDonald, G.A., and Katsura, T., 1964, Chemical composition of Hawaiian lavas: *Journal of Petrology*, v. 5, p. 82-133.
- Miyashiro, A., 1974, Volcanic rock series in island arcs and active continental margins: *American Journal of Science*, v. 274, p. 321-355.
- Parker, A., 1970, An index of weathering for silicate rocks: *Geological Magazine*, v. 107, p. 501-504.
- Rittman, A., 1962 (Vincent, E.A., Translator), *Volcanoes and their activity*: Interscience Publishers, 305 p.
- Segerstrom, K., and Young, E.J., 1972, General geology of the Hahns Peak and Farwell Mountain quadrangles, Routt County, Colorado: U.S. Geological Survey Bulletin 1349.

- Shand, S.J., 1943, Eruptive rocks: John Wiley and Sons, 2d ed.
- Sugimura A., 1968, Spatial relations of basaltic magmas in island arcs, in Hess, H.H., and Poldervaart, A., ed., Basalts: The Poldervaart Treatise on Rocks of Basaltic Composition: Interscience Publishers, v. 2, p. 537-571.
- Zanettin, B., 1984, Proposed new chemical classification of volcanic rocks: Episodes, v. 7, no. 4, p. 19-20.

APPENDIX A. PETCAL (version 2.1) program

```
100 PRINT "REMEMBER TO TURN ON PRINTER FOR OUTPUT"
110 PRINT "REMEMBER TO ACTIVATE 'CAPS LOCK'"
120 DIM W(20),M(20),N(20),Z(30),MS(30)
130 FOR A=1 TO 30
140 READ MS(A)
150 NEXT A
160 DATA QUARTZ, CORUNDUM, OROHCLASE, ALBITE, ANORTHITE
170 DATA LEUCITE, NEPHELINE, KALSILITE, ACMITE
180 DATA SODIUM METASILICATE, POTASSIUM METASILICATE
190 DATA WOLLASTONITE, DIOPSIDE, HYPERTHENE, OLIVINE
200 DATA CALCIUM ORTHOSILICATE, MAGNETITE, HEMATITE
210 DATA ILMENITE, SPHENE, PEROVSKITE, RUTILE, APATITE
220 DATA WOLLASTONITE(DIOPSIDE), ENSTATITE(DIOPSIDE)
230 DATA FERROSILITE(DIOPSIDE), ENSTATITE(HYPERSTHENE)
240 DATA FERROSILITE(HYPERSTHENE), FORSTERITE(OLIVINE)
250 DATA FAYALITE(OLIVINE)
260 PRINT
270 INPUT "IS DATA IN SEPARATE DATAFILE (Y/N) ";A$#
280 IF A$#="N" THEN 310
290 INPUT "ENTER DATAFILE NAME ";AB$
300 OPEN AB$ FOR INPUT AS #1
310 INPUT "SHALL I USE NORMALIZED OXIDE VALUES (Y/N)"; AC$#
320 INPUT "SHALL I RECALCULATE FE2O3-FEO (Y/N)"; AD$#
330 FOR T=1 TO 999
340 IF A$#="N" THEN 370
350 INPUT #1,SNS,W(1),W(2),W(3),W(4),W(5),W(6),W(7),W(8),W(9),W(10),W(11)
360 GOTO 550
370 INPUT "ENTER SAMPLE NUMBER ";SNS#
380 PRINT
390 INPUT "INPUT SiO2      ";W(1)
400 INPUT "INPUT Al2O3    ";W(2)
410 INPUT "INPUT Fe2O3    ";W(3)
420 INPUT "INPUT FeO       ";W(4)
430 INPUT "INPUT MgO       ";W(5)
440 INPUT "INPUT CaO       ";W(6)
450 INPUT "INPUT Na2O     ";W(7)
460 INPUT "INPUT K2O       ";W(8)
470 INPUT "INPUT TiO2     ";W(9)
480 INPUT "INPUT P2O5     ";W(10)
490 INPUT "INPUT MnO      ";W(11)
500 PRINT
510 INPUT "IS DATA CORRECT (Y/N) ";AE$#
520 IF AE$#="Y" GOTO 550
530 PRINT "REENTER DATA"
540 GOTO 380
550 LPRINT
560 LPRINT"$$$$*****$$$$$*****$$$$$*****$$$$$*****$$$$$*****$$$$$*****$$"
570 LPRINT "SAMPLE NUMBER   "SNS#
580 LPRINT
```

```

590 LPRINT "** INPUT DATA **"
600 LPRINT "SIO2";W(1),"AL203";W(2),"FE203";W(3),"FEO";W(4)
610 LPRINT "MGO ";W(5),"CAO ";W(6),"NA2O ";W(7),"K2O";W(8)
620 LPRINT "TIO2";W(9),"P205 ";W(10),"MNO ";W(11)
630 IF AD$="N" THEN 840
640 IF W(1)>=52 THEN 710
650 IF W(3)=0 THEN 670
660 IF W(3)/(W(3)+W(4))=<.2 THEN 840
670 B1=W(4)
680 W(4)=.816*W(4)+.735*W(3)
690 W(3)=.204*B1+.184*W(3)
700 GOTO 820
710 IF W(3)=0 THEN 760
720 IF W(3)<(W(9)+1.5 THEN 840
730 W(4)=.8998*(W(3)-W(9)-1.5)+W(4)
740 W(3)= W(9)+1.5
750 GOTO 820
760 W(3)=W(9)+1.5
770 IF W(4)>W(3)*.8998 THEN 810
780 W(3)=0
790 LPRINT "$$ NORMS ARE CALCULATED ASSUMING NO FE203 $$"
800 GOTO 820
810 W(4)=W(4)-.8998*W(3)
820 LPRINT "MODIFIED FE203= ";W(3)
830 LPRINT "MODIFIED FEO= ";W(4)
840 B2= W(1)+W(2)+W(3)+W(4)+W(5)+W(6)+W(7)+W(8)+W(9)+W(10)+W(11)
850 LPRINT "SUM OF OXIDES= ";B2
860 IF ACS = "N" THEN 940
870 LPRINT "** NORMALIZED OXIDE VALUES **"
880 FOR J=1 TO 11
890 W(J)=W(J)*100/B2
900 NEXT J
910 LPRINT "SIO2";W(1),"AL203";W(2),"FE203";W(3),"FEO";W(4)
920 LPRINT "MGO ";W(5),"CAO ";W(6),"NA2O ";W(7),"K2O";W(8)
930 LPRINT "TIO2";W(9),"P205 ";W(10),"MNO ";W(11)
940 W(15)=.8998*W(3)+W(4)
950 LPRINT
960 LPRINT "** MOLE NUMBERS **"
970 M(1)=W(1)/60.08
980 M(2)=W(2)/101.96
990 M(3)=W(3)/159.69
1000 M(4)=W(4)/71.85
1010 M(5)=W(5)/40.31
1020 M(6)=W(6)/56.08
1030 M(7)=W(7)/61.98
1040 M(8)=W(8)/94.2
1050 M(9)=W(9)/79.9
1060 M(10)=W(10)/141.94
1070 M(11)=W(11)/70.94
1080 M(15)=W(15)/71.85
1090 LPRINT "SI";M(1),"AL";M(2),"FE+3";M(3),"FE+2";M(4)
1100 LPRINT "MG";M(5),"CA";M(6),"NA";M(7),"K";(8)
1110 LPRINT "TI";M(9),"P";M(10),"MN";M(11)
1120 LPRINT

```

```

1130 LPRINT "** NIGGLI NUMBERS **"
1140 B3=M(2)+M(5)+M(6)+M(7)+M(8)+M(11)+M(15)
1150 LPRINT "AL";(M(2)*100/B3),"FM ";(M(15)+M(11)+M(5))*100/B3
1160 LPRINT "C ";M(6)*100/B3,"ALK";(M(7)+M(8))*100/B3,"SI";M(1)*100/B3
1170 LPRINT "K ";M(8)/(M(7)+M(8)),"MG ";M(5)/(M(15)+M(11)+M(5))
1180 LPRINT "TI";M(9)*100/B3,"P ";M(10)*100/B3
1190 LPRINT
1200 LPRINT "** OXIDE-SILICA RATIOS **"
1210 LPRINT "AL2O3/SIO2      ";W(2)/W(1)
1220 LPRINT "FE2O3/SIO2      ";W(3)/W(1)
1230 LPRINT "FEO/SIO2        ";W(4)/W(1)
1240 LPRINT "FEO*/SIO2       ";W(15)/W(1)
1250 LPRINT "MGO/SIO2        ";W(5)/W(1)
1260 LPRINT "CAO/SIO2         ";W(6)/W(1)
1270 LPRINT "NA2O/SIO2        ";W(7)/W(1)
1280 LPRINT "K2O/SIO2         ";W(8)/W(1)
1290 LPRINT
1300 LPRINT "** OTHER OXIDE RATIOS **"
1310 LPRINT "NA2O/K2O          ";W(7)/W(8)
1320 LPRINT "K2O/NA2O          ";W(8)/W(7)
1330 LPRINT "FEO*               ";W(15)
1340 LPRINT "FEO*/MGO          ";W(15)/W(5)
1350 LPRINT "NA2O + K2O        ";W(7)+W(8)
1360 B5=(W(7)+W(8)+W(5)+W(15))/100
1370 LPRINT "A:F:M=           ";(W(7)+W(8))/B5;W(15)/B5;W(5)/B5
1380 B6=(W(7)+W(8)+W(6))/100
1390 LPRINT "NA2O:K2O:CAO       ";W(7)/B6;W(8)/B6;W(6)/B6
1400 LPRINT
1410 LPRINT "MGO:AL2O3:(CAO+NA2O+K2O)   (MOLE PROP.)"
1420 B7=(M(2)+M(5)+M(6)+M(7)+M(8))/100
1430 LPRINT M(5)/B7;M(2)/B7;(M(6)+M(7)+M(8))/B7
1440 LPRINT
1450 LPRINT
1460 LPRINT "**** NORMATIVE MINERALS ****"
1470 FOR K=1 TO 23
1480 Z(K)=0
1490 NEXT K
1500 FOR J=1 TO 15
1510 N(J)=M(J)
1520 NEXT J
1530 N(16)=N(6)
1540 N(4)=N(4)+N(11)
1550 IF N(10)>0 THEN 1570
1560 GOTO 1640
1570 Z(23)=N(10)
1580 N(6)=N(6)-3.33*N(10)
1590 IF N(6)>=0 THEN 1640
1600 Z(23)=N(16)
1610 N(10)=N(10)-N(16)/3.33
1620 N(6)=0
1630 LPRINT " ROCK CONTAINS EXCESS P2O5 OVER CAO ";N(10)*141.94

```

1640 IF N(4)<N(9) THEN 1690
1650 Z(19)=N(9)
1660 N(4)=N(4)-N(9)
1670 N(9)=0
1680 GOTO 1720
1690 Z(19)=N(4)
1700 N(9)=N(9)-N(4)
1710 N(4)=0
1720 IF N(2)<N(8) THEN 1760
1730 Z(3)=N(8)
1740 N(2)=N(2)-N(8)
1750 GOTO 1810
1760 Z(3)=N(2)
1770 Z(11)=N(8)-N(2)
1780 N(2)=0
1790 N(8)=0
1800 GOTO 1970
1810 IF N(2)<N(7) THEN 1860
1820 Z(4)=N(7)
1830 N(2)=N(2)-N(7)
1840 N(7)=0
1850 GOTO 1890
1860 Z(4)=N(2)
1870 N(7)=N(7)-N(2)
1880 GOTO 1970
1890 IF N(2)<N(6) THEN 1950
1900 Z(5)=N(6)
1910 N(2)=N(2)-N(6)
1920 Z(2)=N(2)
1930 N(6)=0
1940 GOTO 1970
1950 Z(5)=N(2)
1960 N(6)=N(6)-N(2)
1970 IF N(9)<N(6) THEN 2030
1980 Z(20)=N(6)
1990 N(9)=N(9)-N(6)
2000 Z(22)=N(9)
2010 N(6)=0
2020 GOTO 2060
2030 Z(20)=N(9)
2040 N(6)=N(6)-N(9)
2050 N(9)=0
2060 IF N(3)<N(7) THEN 2100
2070 Z(9)=N(7)
2080 N(3)=N(3)-N(7)
2090 GOTO 2130
2100 Z(9)=N(3)
2110 Z(10)=N(7)-N(3)
2120 N(3)=0
2130 IF N(4)<N(3) THEN 2170
2140 Z(17)=N(3)
2150 N(4)=N(4)-N(3)
2160 GOTO 2210

2170 Z(17)=N(4)
2180 Z(18)=N(3)-N(4)
2190 N(3)=N(3)-N(4)
2200 N(4)=0
2210 D=N(5)+N(4)
2220 IF D=0 THEN 2340
2230 D1=N(5)/D
2240 D2=N(4)/D
2250 IF N(6)>D THEN 2320
2260 Z(13)=N(6)
2270 D=D-N(6)
2280 N(5)=D*D1
2290 N(4)=D*D2
2300 Z(14)=D
2310 GOTO 2350
2320 Z(13)=D
2330 N(6)=N(6)-D
2340 Z(12)=N(6)
2350 D3=Z(20)+4*Z(9)+Z(10)+Z(11)+6*Z(3)+6*Z(4)+2*Z(5)+2*Z(13)+Z(14)+Z(12)
2360 D4=N(1)-D3
2370 IF D4<0 THEN 2400
2380 Z(1)=D4
2390 GOTO 2930
2400 D4=ABS(D4)
2410 IF D4<Z(14)/2 THEN 2460
2420 Z(15)=Z(14)/2
2430 D5=D4-Z(14)/2
2440 Z(14)=0
2450 GOTO 2490
2460 Z(15)=D4
2470 Z(14)=Z(14)-2*D4
2480 GOTO 2930
2490 IF D5<Z(20) THEN 2540
2500 Z(21)=Z(20)
2510 D6=D5-Z(20)
2520 Z(20)=0
2530 GOTO 2570
2540 Z(20)=Z(20)-D5
2550 Z(21)=D5
2560 GOTO 2930
2570 IF D6<4*Z(4) THEN 2620
2580 Z(7)=Z(4)
2590 D7=D6-4*Z(4)
2600 Z(4)=0
2610 GOTO 2650
2620 Z(7)=D6/4
2630 Z(4)=Z(4)-D6/4
2640 GOTO 2930
2650 IF D7<2*Z(3) THEN 2700
2660 Z(6)=Z(3)
2670 D8=D7-2*Z(3)
2680 Z(3)=0
2690 GOTO 2730

```
2700 Z(6)=D7/2
2710 Z(3)=Z(3)-D7/2
2720 GOTO 2930
2730 IF D8<Z(12)/2 THEN 2780
2740 Z(16)=Z(12)/2
2750 D9=D8-Z(12)/2
2760 Z(12)=0
2770 GOTO 2810
2780 Z(16)=D8
2790 Z(12)=Z(12)-2*D8
2800 GOTO 2930
2810 IF D9<Z(13) THEN 2870
2820 Z(16)=Z(16)+Z(13)/2
2830 Z(15)=Z(15)+Z(13)/2
2840 E1=D9-Z(13)
2850 Z(13)=0
2860 GOTO 2910
2870 Z(16)=Z(16)+D9/2
2880 Z(15)=Z(15)+D9/2
2890 Z(13)=Z(13)-D9
2900 GOTO 2930
2910 Z(8)=E1/2
2920 Z(6)=Z(6)-E1/2
2930 LPRINT
2940 E2=Z(13)
2950 E3=Z(14)
2960 E4=Z(15)
2970 Z(1)=Z(1)*60.08
2980 Z(2)=Z(2)*101.96
2990 Z(3)=Z(3)*556.64
3000 Z(4)=Z(4)*524.42
3010 Z(5)=Z(5)*278.2
3020 Z(6)=Z(6)*436.48
3030 Z(7)=Z(7)*284.1
3040 Z(8)=Z(8)*316.32
3050 Z(9)=Z(9)*461.99
3060 Z(10)=Z(10)*122.06
3070 Z(11)=Z(11)*154.28
3080 Z(12)=Z(12)*116.16
3090 Z(13)=(Z(13)*216.55*D1)+(Z(13)*248.09*D2)
3100 Z(14)=(Z(14)*D1)*100.39+(Z(14)*D2)*131.93
3110 Z(15)=(Z(15)*D1)*140.7+(Z(15)*D2)*203.78
3120 Z(16)=Z(16)*172.24
3130 Z(17)=Z(17)*231.54
3140 Z(18)=Z(18)*159.69
3150 Z(19)=Z(19)*151.75
3160 Z(20)=Z(20)*196.06
3170 Z(21)=Z(21)*135.98
3180 Z(22)=Z(22)*79.9
3190 Z(23)=Z(23)*328.67
3200 Z(24)=E2*116.16
3210 Z(25)=E2*D1*100.39
```

```

3220 Z(26)=E2*D2*131.93
3230 Z(27)=E3*D1*100.39
3240 Z(28)=E3*D2*131.93
3250 Z(29)=E4*D1*140.7
3260 Z(30)=E4*D2*203.78
3270 FOR K=1 TO 23
3280 A=K
3290 IF Z(K)=0 THEN 3310
3300 LPRINT M$(A),Z(K)
3310 NEXT K
3320 U=0
3330 FOR K=1 TO 23
3340 U=U+Z(K)
3350 NEXT K
3360 LPRINT "* SUM *",U
3370 LPRINT
3380 FOR K=24 TO 30
3390 A=K
3400 IF Z(K)<=8.999999E-03 THEN 3420
3410 LPRINT M$(A),Z(K)
3420 NEXT K
3430 LPRINT
3440 LPRINT "** NORMATIVE RATIOS - CIPW **"
3450 F1=Z(3)+Z(4)+Z(5)
3460 IF F1=0 THEN 4400
3470 F2=F1/100
3480 LPRINT "OR:AB:AN   ";Z(3)/F2;Z(4)/F2;Z(5)/F2
3490 F3=(Z(1)+Z(3)+Z(4))/100
3500 LPRINT "Q:OR:AB   ";Z(1)/F3;Z(3)/F3;Z(4)/F3
3510 F4=(Z(1)+Z(3)+Z(4)+Z(5))/100
3520 LPRINT "Q:OR:AB:AN   ";Z(1)/F4;Z(3)/F4;(Z(4)+Z(5))/F4
3530 F5=Z(6)+Z(7)+Z(8)
3540 F6=(F1+F5)/100
3550 LPRINT "LC+NE+KS:OR:AB:AN   ";F5/F6;Z(3)/F6;(Z(4)+Z(5))/F6
3560 IF (Z(4)+Z(5)+(5*Z(7)/3))=0 THEN 4370
3570 F7=(100*Z(5))/(Z(4)+Z(5)+(5*Z(7)/3))
3580 LPRINT "NORMATIVE PLAGIOCLASE CONTENT= AN ";F7
3590 LPRINT "NORMATIVE COLOR INDEX=   ";Z(15)+Z(14)+Z(13)+Z(19)+Z(18)+Z(17)
3600 LPRINT
3610 LPRINT "** PETROCHEMICAL INDICES **"
3620 LPRINT "ALKALI INDEX   ";(W(8)*100)/(W(7)+W(8))
3630 LPRINT "FELSIC INDEX   ";(W(7)+W(8))*100/(W(6)+W(7)+W(8))
3640 LPRINT "MAFIC INDEX   ";(W(3)+W(4))*100/(W(3)+W(4)+W(5))
3650 F8=(W(2)+W(6)+W(7)+W(8))/(W(2)+W(6)-(W(7)+W(8)))
3660 LPRINT "ALKALINITY RATIO   ";F8
3670 LPRINT "BASICITY INDEX   ";W(3)+W(4)+.5*(W(6)+W(5))
3680 LPRINT "SOLIDIFICATION INDEX   ";(W(5)*100)/(W(5)+W(15)+W(7)+W(8))
3690 G1=Z(5)+2.157*Z(25)+Z(29)+.70084*Z(27)
3700 LPRINT "CRYSTALLIZATION INDEX   ";G1
3710 LPRINT "DIFFERENTIATION INDEX   ";Z(1)+Z(3)+Z(4)+Z(7)+Z(6)+Z(8)
3720 LPRINT "FELSIC-MAFIC INDEX   ";(W(1)+W(7)+W(8))/(W(3)+W(4)+W(5)+W(6))
3730 G2=100*(.02756*W(5)+.02547*W(6)+9.213999E-02*W(7)+.08493*W(8))
3740 LPRINT "WEATHERING INDEX   ";G2

```

```

3750 G3=(W(5)+W(8))*100/(W(5)+W(6)+W(7)+W(8))
3760 LPRINT "ALTERATION INDEX          ";G3
3770 LPRINT "PERALUMINOUS INDEX      ";100*(M(2)-M(6)-M(7)-M(8))/M(2)
3780 LPRINT "PERALKALINE INDEX        ";(M(7)+M(8))/M(2)
3790 LPRINT "THETA INDEX             ";W(1)-47*(W(7)+W(8))/W(2)
3800 LPRINT "SUITE INDEX              ";(W(7)+W(8))^2/(W(1)-43)
3810 LPRINT
3820 IF M(2)>M(6)+M(7)+M(8) THEN 3870
3830 IF M(2)>M(7)+M(8)+.05*(M(7)+M(8)) THEN 3890
3840 IF M(2)>M(7)+M(8)-.05*(M(7)+M(8)) THEN 3910
3850 LPRINT "THE ROCK IS PERALKALINE (SHAND,1945)"
3860 GOTO 3920
3870 LPRINT "THE ROCK IS PERALUMINOUS (SHAND,1945)"
3880 GOTO 3920
3890 LPRINT "THE ROCK IS METALUMINOUS (SHAND,1945)"
3900 GOTO 3920
3910 LPRINT "THE ROCK IS SUBALUMINOUS (SHAND,1945)"
3920 G6=W(7)+W(8)
3930 G7=(-3.3539E-04)*(G6^6)+(.01203)*(G6^5)-(.15188)*(G6^4)
3940 G7=G7+(.86096)*(G6^3)-2.111*(G6^2)+(3.9492*G6)+39
3950 IF W(1)>=G7 THEN 3980
3960 LPRINT "THE ROCK IS ALKALIC      (IRVINE & BARAGAR,1971)"
3970 GOTO 4100
3980 LPRINT "THE ROCK IS SUBALKALIC (IRVINE & BARAGAR,1971)"
3990 IF F7<40 THEN 4020
4000 IF W(2)>=12+.08*F7 THEN 4070
4010 GOTO 4090
4020 G8=W(5)/((G6+W(15))+W(5))/100
4030 G9=(1.5559E-12)*(G8^8)-(7.7142E-10)*(G8^7)+(1.5664E-07)*(G8^6)
4040 G9=G9-(1.6738E-05)*(G8^5)+(.0010017)*(G8^4)-(.032552)*(G8^3)
4050 G9=G9+(.47776)*(G8^2)-(1.1085)*(G8)+30
4060 IF (W(15)/((G6+W(15))+W(5))/100))>=G9 THEN 4090
4070 LPRINT "THE ROCK IS CALC-ALKALIC (IRVINE & BARAGAR,1971)"
4080 GOTO 4100
4090 LPRINT "THE ROCK IS THOLEIITIC (IRVINE & BARAGAR,1971)"
4100 IF W(1)>=6.4*((W(15))/W(5))+42.8 THEN 4130
4110 LPRINT "THE ROCK IS THOLEIITIC (MIYASHIRO,1974)"
4120 GOTO 4140
4130 LPRINT "THE ROCK IS CALC-ALKALIC (MIYASHIRO,1974)"
4140 IF W(1)>57.5 THEN 4180
4150 IF W(1)<3.047*(W(7)+W(8))+38.3 THEN 4270
4160 IF W(1)<5.917*(W(7)+W(8))+37.4 THEN 4250
4170 GOTO 4230
4180 IF W(1)<9.26*(W(7)+W(8))-3.8 THEN 4270
4190 IF W(1)>60 THEN 4220
4200 IF W(1)<5.917*(W(7)+W(8))+37.4 THEN 4250
4210 GOTO 4230
4220 IF W(1)<6.667*(W(7)+W(8))+25.6 THEN 4250
4230 LPRINT "THE ROCK IS THOLEIITIC (KUNO,1966)"
4240 GOTO 4280
4250 LPRINT "THE ROCK IS HIGH ALUMINA (KUNO,1966)"
4260 GOTO 4280
4270 LPRINT "THE ROCK IS ALKALIC (KUNO,1966)"

```

```
4280 IF W(1)>=2.71*(W(7)+W(8))+39 THEN 4310
4290 LPRINT "THE ROCK IS ALKALIC      (MACDONALD & KATSURA,1964)"
4300 GOTO 4330
4310 LPRINT "THE ROCK IS THOLEIITIC   (MACDONALD & KATSURA,1964)"
4320 IF (M(2)/(M(7)+M(8)+M(6)))>1.1 THEN 4350
4330 LPRINT "THE ROCK IS PROBABLY I-TYPE (CHAPPELL & WHITE,1974)"
4340 GOTO 4360
4350 LPRINT "THE ROCK IS PROBABLY S-TYPE (CHAPPELL & WHITE,1974)"
4360 GOTO 4390
4370 LPRINT
4380 LPRINT "$$ THIS ROCK IS ATYPICAL - NO PLAG OR NEPHELINE $$"
4390 IF Z(3)>0 THEN 4410
4400 LPRINT "$$ THIS ROCK IS ATYPICAL - NO PLAG OR K-SPAR $$"
4410 LPRINT
4420 LPRINT "ACCORDING TO THE IUGS CLASSIFICATION (ZANETTIN,1984)"
4430 H1=W(1)*100/B2
4440 H2=(W(7)+W(8))*100/B2
4450 H3=Z(15)*100/U
4460 IF H1>45 THEN 4550
4470 IF H2>3 THEN 4500
4480 LPRINT "THE ROCK IS A PICROBASALT"
4490 GOTO 5080
4500 IF H3>10 THEN 4530
4510 LPRINT "THE ROCK IS A TEPHRITE"
4520 GOTO 5080
4530 LPRINT "THE ROCK IS A BASANITE"
4540 GOTO 5080
4550 IF H2>5 THEN 4590
4560 IF H1>52 THEN 4590
4570 LPRINT "THE ROCK IS A BASALT"
4580 GOTO 5080
4590 IF H1<4*H2+32.2 THEN 4740
4600 IF H1>57 THEN 4630
4610 LPRINT "THE ROCK IS A BASALTIC ANDESITE"
4620 GOTO 5080
4630 IF H1>-.25974*H2+65 THEN 4660
4640 LPRINT "THE ROCK IS AN ANDESITE"
4650 GOTO 5080
4660 IF H1>-1*H2+77 THEN 4690
4670 LPRINT "THE ROCK IS A DACITE"
4680 GOTO 5080
4690 IF (M(7)+M(8))/M(2)>1 THEN 4720
4700 LPRINT "THE ROCK IS A RHYOLITE"
4710 GOTO 5080
4720 LPRINT "THE ROCK IS AN ALKALI RHYOLITE"
4730 GOTO 5080
4740 IF H1<1.88235*H2+35.58235 THEN 4950
4750 IF H1>-1.31818*H2+58.59 THEN 4780
4760 LPRINT "THE ROCK IS A TRACHYBASALT"
4770 GOTO 5080
4780 IF H1>-1.36667*H2+65.7 THEN 4810
4790 LPRINT "THE ROCK IS A BASALTIC TRACHYANDESITE"
4800 GOTO 5080
```

```
4810 IF H1>-1.35*H2+73.395 THEN 4840
4820 LPRINT "THE ROCK IS A TRACHYANDESITE"
4830 GOTO 5080
4840 IF H1>1.66667*H2+56.41667 THEN 4900
4850 IF (M(7)+M(8))/M(2)>1 THEN 4880
4860 LPRINT "THE ROCK IS A RHYOLITE"
4870 GOTO 5080
4880 LPRINT "THE ROCK IS AN ALKALI RHYOLITE"
4890 GOTO 5080
4900 IF (M(7)+M(8))/M(2)>1 THEN 4930
4910 LPRINT "THE ROCK IS A TRACHYTE"
4920 GOTO 5080
4930 LPRINT "THE ROCK IS AN ALKALI TRACHYTE"
4940 GOTO 5080
4950 IF H1>-2.25*H2+65.3 THEN 5010
4960 IF H3>10 THEN 4990
4970 LPRINT "THE ROCK IS A TEPHRITE"
4980 GOTO 5080
4990 LPRINT "THE ROCK IS A BASANITE"
5000 GOTO 5080
5010 IF H1>-2.13636*H2+72.5545 THEN 5040
5020 LPRINT "THE ROCK IS A PHONOTEPHRITE"
5030 GOTO 5080
5040 IF H1>-2.21739*H2+83.54348 THEN 5070
5050 LPRINT "THE ROCK IS A TEPHRIPHONOLITE"
5060 GOTO 5080
5070 LPRINT "THE ROCK IS A PHONOLITE"
5080 LPRINT
5090 IF A$="Y" THEN 5130
5100 PRINT
5110 INPUT "ANOTHER SAMPLE (Y/N) ";AF$
5120 IF AF$="N" THEN 5150
5130 NEXT T
5140 CLOSE #1
5150 STOP
5160 END
```

APPENDIX B. Sample Output

SAMPLE NUMBER ROCK SAMPLE 1

** INPUT DATA **

SiO2 55.8	Al2O3 18.1	Fe2O3 6.9	FeO 0
MgO 3.1	CaO 9.8	Na2O 3.5	K2O 1.5
TiO2 .9	P2O5 9.000001E-02		MnO .17

MODIFIED Fe2O3= 2.4

MODIFIED FeO= 4.0491

SUM OF OXIDES= 99.40911

** NORMALIZED OXIDE VALUES **

SiO2 56.13168	Al2O3 18.20759	Fe2O3 2.414266
FeO 4.073168		
MgO 3.118427	CaO 9.858252	Na2O 3.520805
K2O 1.508916		
TiO2 .9053497	P2O5 9.053497E-02	MnO .1710105

** MOLE NUMBERS **

Si .9342822	Al .1785758	Fe+3 1.511845E-02	Fe+2 5.668989E-02
Mg 7.736111E-02		Ca .1757891	Na 5.680549E-02
Ti 1.133104E-02		P 6.378397E-04	K 8
			Mn 2.410636E-03

** NIGGLI NUMBERS **

Al 30.0691	FM 28.06878
C 29.59986	ALK 12.26226
K .2199589	SI 157.3171
Ti 1.907952	P .1074013

** OXIDE-SILICA RATIOS **

Al2O3/SiO2	.3243728
Fe2O3/SiO2	4.301076E-02
FeO/SiO2	7.256452E-02
FeO*/SiO2	.1112656
MgO/SiO2	5.555555E-02
CaO/SiO2	.1756272
Na2O/SiO2	6.272402E-02
K2O/SiO2	2.688172E-02

** OTHER OXIDE RATIOS **

Na2O/K2O	2.333334
K2O/Na2O	.4285714
FeO*	6.245525
FeO*/MgO	2.002781
Na2O + K2O	5.029721
A:F:M=	34.94397 43.39077 21.66526
Na2O:K2O:CaO	23.64865 10.13514 66.21622

MgO:Al2O3:(CaO+Na2O+K2O) (MOLE PROP.)

15.33271 35.39311 49.2742

**** NORMATIVE MINERALS ****

QUARTZ	6.483271
ORTHOCLASE	8.916382
ALBITE	29.78994
ANORTHITE	29.42023
DIOPSIDE	15.34228
HYPERTHENE	4.620422
MAGNITITE	3.500527
ILMENITE	1.719485
APATITE	.2096388
* SUM *	100.0022

WOLLASTONITE(DIOPSIDE)	7.888772
ENSTATITE(DIOPSIDE)	4.794301
FERROSILITE(DIOPSIDE)	2.65921
ENSTATITE(HYPERSTHENE)	2.971981
FERROSILITE(HYPERSTHENE)	1.648441

** NORMATIVE RATIOS - CIPW **

OR:AB:AN	13.08797	43.72736	43.18469
Q:OR:AB	14.34683	19.73105	65.92212
Q:OR:AB+AN	8.689569	11.95068	79.35976
LC+NE+KS:OR:AB+AN	0	13.08797	86.91204
NORMATIVE PLAGIOCLASE CONTENT = AN	49.6878		
NORMATIVE COLOR INDEX =	25.18272		

** PETROCHEMICAL INDICES **

ALKALI INDEX	30
FELSIC INDEX	33.78378
MAFIC INDEX	67.53621
ALKALINITY RATIO	1.436681
BASICITY INDEX	12.97577
SOLIDIFICATION INDEX	21.66526
CRYSTALLIZATION INDEX	41.84442
DIFFERENTIATION INDEX	45.18959
FELSIC-MAFIC INDEX	3.142265
WEATHERING INDEX	78.95926
ALTERATION INDEX	25.69832
PERALUMINOUS INDEX	-39.21975
PERALKALINE INDEX	.4078028
THETA INDEX	43.14826
SUITE INDEX	1.926493

THE ROCK IS METALUMINOUS (SHAND, 1945)
 THE ROCK IS SUBALKALIC (IRVINE & BARAGAR, 1971)
 THE ROCK IS CALC-ALKALIC (IRVINE & BARAGAR, 1971)
 THE ROCK IS CALC-ALKALIC (MIYASHIRO, 1974)
 THE ROCK IS HIGH ALUMINA (KUNO, 1966)
 THE ROCK IS THOLEIITIC (MACDONALD & KATSURA, 1964)
 THE ROCK IS PROBABLY I-TYPE (CHAPPELL & WHITE, 1974)

ACCORDING TO THE IUGS CLASSIFICATION (ZANETTIN, 1984)
 THE ROCK IS A BASALTIC ANDESITE

APPENDIX C. PETDAT Program

PETDAT (PETrologic DATa) is a BASIC language program for creating a datafile compatible with PETCAL. The program offers the convenience of being able to append an existing datafile as well as creating a new datafile.

After loading PETDAT under BASIC, enter the name of the datafile to be created (or appended). The name should be limited to eight characters. If a preexisting datafile created by PETDAT is to be added to, indicate "Y" after the second prompt. Then enter the sample name which may be of any length. The data is then entered one oxide value at a time. After the data is entered, the program will ask if it is correct. If not, all of the data must be reentered. The program will then ask if there are more samples to be entered. If "Y", the program will recycle to enter another sample. If "N", the program will terminate. The data will then be ready to be called into the PETCAL program.

```

100 PRINT "PETDAT - A program to create a datafile for PETCAL"
110 PRINT
120 DIM W(11)
130 T=1
140 INPUT "ENTER NAME OF DATAFILE ";PNS
150 AAS=PNS
160 INPUT "IS THIS AN EXISTING FILE BEING ADDED TO (Y/N) ";ABS
170 IF ABS="N" THEN 190
180 T=T+1
190 PRINT
200 INPUT "ENTER SAMPLE NAME ";SNS
210 INPUT "ENTER SiO2      ";W(1)
220 INPUT "ENTER Al2O3     ";W(2)
230 INPUT "ENTER Fe2O3     ";W(3)
240 INPUT "ENTER FeO       ";W(4)
250 INPUT "ENTER MGO       ";W(5)
260 INPUT "ENTER CAO       ";W(6)
270 INPUT "ENTER Na2O      ";W(7)
280 INPUT "ENTER K2O       ";W(8)
290 INPUT "ENTER TiO2      ";W(9)
300 INPUT "ENTER P2O5      ";W(10)
310 INPUT "ENTER MnO      ";W(11)
320 PRINT
330 INPUT "IS THE DATA CORRECT (Y/N) ";ACS
340 IF ACS="Y" THEN 370
350 PRINT "REENTER DATA"
360 GOTO 210
370 IF T>1 THEN 420
380 OPEN AAS FOR OUTPUT AS #1
390 WRITE #1,SNS,W(1),W(2),W(3),W(4),W(5),W(6),W(7),W(8),W(9),W(10),W(11)
400 CLOSE #1
410 GOTO 450
420 OPEN AAS FOR APPEND AS #1
430 WRITE #1,SNS,W(1),W(2),W(3),W(4),W(5),W(6),W(7),W(8),W(9),W(10),W(11)
440 CLOSE #1
450 PRINT
460 INPUT "ANYMORE SAMPLES TO ENTER (Y/N) ";ADS
470 IF ADS="Y" THEN 180
480 PRINT
490 END

```